



Trading places: China, ASEAN, and the 2025 tariffs

STEP 1 Get and prepare the data

Download monthly US imports (\$ millions, Census data) from China and ASEAN's 11 members, then prep:

```
import fred IMPCH IMP5520 IMP5490 IMP5600 IMP5570 IMP5650 ///
IMP5590 IMP5550 IMP5610 IMP5460 IMP5530 IMP5601, clear†
gen mdate = mofd(daten)
format mdate %tm
replace IMP5601 = 0 if IMP5601 == .
gen asean = IMP5520 + IMP5490 + IMP5600 + IMP5570 + IMP5650 + ///
IMP5590 + IMP5550 + IMP5610 + IMP5460 + IMP5530 + IMP5601
rename IMPCH china
```

STEP 2 Total the past 12 months, then picture it

Add up the current month and the previous eleven, then plot from 2022 with labels:

```
sort mdate
gen china12 = (china + china[_n-1] + china[_n-2] + china[_n-3] + ///
china[_n-4] + china[_n-5] + china[_n-6] + china[_n-7] + ///
china[_n-8] + china[_n-9] + china[_n-10] + china[_n-11]) / 1000
gen asean12 = (asean + asean[_n-1] + asean[_n-2] + asean[_n-3] + ///
asean[_n-4] + asean[_n-5] + asean[_n-6] + asean[_n-7] + ///
asean[_n-8] + asean[_n-9] + asean[_n-10] + asean[_n-11]) / 1000
keep if mdate >= tm(2000m1)
tway line china12 asean12 mdate if mdate >= tm(2022m1), legend(off) ///
xline(`=tm(2025m1)', lpattern(dash)) xscale(range(. `=tm(2027m8)')) ///
text(560 `=tm(2025m1)' "Start of Trump term", place(w) size(small)) ///
text(264 `=tm(2026m5)' "U.S. imports" "from China", place(e) size(small)) ///
text(512 `=tm(2026m5)' "U.S. imports from" "ASEAN nations", place(e) size(small)) ///
name(rerouting, replace) xsize(8) ysize(4.5)
```

Latest 12-month totals — China: \$ _____ b. ASEAN: \$ _____ b. ASEAN passes China in: _____

STEP 3 Integration or substitution?

Scatter ASEAN against China in each era, then estimate each slope:

```
tway (scatter asean12 china12) (lfit asean12 china12) if mdate < tm(2025m1), legend(off) ///
title("2000-2024: Before the trade war", size(medium)) ///
subtitle("Imports from ASEAN countries tended to rise" ///
"with imports from China", size(medsmall)) name(before, replace)
tway (scatter asean12 china12) (lfit asean12 china12) if mdate >= tm(2025m1), legend(off) ///
title("Since 2025: The trade-war era", size(medium)) ///
subtitle("...but since the trade war began, imports from" ///
"ASEAN rose when imports from China fell", size(medsmall)) name(after, replace)
graph combine before after, ycommon iscale(1) name(substitutes, replace) xsize(8) ysize(4.5)
regress asean12 china12 if mdate < tm(2025m1)
regress asean12 china12 if mdate >= tm(2025m1)
```

Slope, 2000–2024: _____ Slope, since 2025: _____

What do you notice?

1. The scatter's slope flips sign after the tariffs. Tell the economic story of each era in a sentence.

2. Can these data distinguish ASEAN-made goods from rerouted Chinese goods? What data would settle it?

[†] Step 1 needs a free FRED API key: get one at fredaccount.stlouisfed.org/apikeys, then type in Stata:
set fredkey YOUR_KEY, permanently (import fred requires Stata 15.1 or later.)